Rozklad mnohočlenu na součin ireducibilních činitelů

```
> restart;
```

PŘÍKLAD 1: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $x^4 - x^3 - x^2 - x - 2$.

```
> restart;
> f1:=x^4-x^3-x^2-x-2;
```

$$f1 := x^4 - x^3 - x^2 - x - 2$$

Můžeme použít příkaz **factor**:

Rozklad v Q[x]

```
> factor(f1);
```

$$(x-2)\,(x+1)\,(x^2+1)$$

Rozklad v R[x]:

```
> factor(f1,real);
```

$$(x+1.)\,(x-2.)\,(x^2+1.)$$

Rozklad v C[x]

```
> factor(f1,complex);
```

$$(x+1.)\,(x+1.\,I)\,(x-1.\,I)\,(x-2.)$$

Chceme-li se vyhnout zápisu koeficeintů výsledných kořenových činitelů desetinnými čísly, použijeme raději příkaz **split** z balíčku příkazů **polytools**, který kombinujeme s příkazem **convert**.

```
> Rozklad_f1:=polytools[split](f1,x);
```

$$Rozklad\_f1 := (x+1)\,(x-\mathrm{RootOf}(\_Z^2+1))\,(\mathrm{RootOf}(\_Z^2+1)+x)\,(x-2)$$

```
> convert(Rozklad_f1,radical);
```

$$(x+1)\,(x-I)\,(I+x)\,(x-2)$$

Grafické znázornění:

```
> plot(f1,x=-3..3,y=-5..5);
```



PŘÍKLAD 2: Rozložte na součin ireducibilních činitelů v R[x] a C[x] mnohočlen $x^4 - x^2 - 2$.

```
> restart;
> f2:=x^4-x^2-2;
```

$$f2 := x^4 - x^2 - 2$$

Příkaz **factor**:

Rozklad v Q[x]

```
> factor(f2);
```

$$(x^2 - 2)\,(x^2 + 1)$$

Rozklad v R[x]

```
> factor(f2,real);
```

$$(x + 1.414213562)\,(x - 1.414213562)\,(x^2 + 1.)$$

Rozklad v C[x]

```
> factor(f2,complex);
```

$$(x + 1.414213562)\,(x + 1.\,I)\,(x - 1.\,I)\,(x - 1.414213562)$$

Příkaz **polytools[split]**:

```
> Rozklad_f2:=polytools[split](f2,x);
```

$$Rozklad\_f2 := (\mathrm{RootOf}(\_Z^2 + 1) + x)\,(x - \mathrm{RootOf}(\_Z^2 + 1))\,(\mathrm{RootOf}(\_Z^2 - 2) + x)\,(x - \mathrm{RootOf}(\_Z^2 - 2))$$

```
> convert(Rozklad_f2,radical);
```

$$(I + x)\,(x - I)\,(\sqrt{2} + x)\,(x - \sqrt{2})$$

Graf:

```
> plot(f2,x=-2..2,y=-3..2);
```



PŘÍKLAD 3: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $x^4 + 1$

```
> restart;
> f3:=x^4+1;
```

$$f3 := x^4 + 1$$

```
> factor(f3);
```

$$x^4 + 1$$

```
> factor(f3,complex);
```

$$(x + 0.7071067812 + 0.7071067812\,I)\,(x + 0.7071067812 - 0.7071067812\,I)\,(x - 0.7071067812 + 0.7071067812\,I)\,(x - 0.7071067812 - 0.7071067812\,I)$$

```
> Rozklad_f3:=polytools[split](f3,x);
```

$$Rozklad\_f3 := (x - \mathrm{RootOf}(\_Z^4 + 1))\,(x + \mathrm{RootOf}(\_Z^4 + 1)^3)\,(x - \mathrm{RootOf}(\_Z^4 + 1)^3)\,(x + \mathrm{RootOf}(\_Z^4 + 1))$$

```
> convert(Rozklad_f3,radical);
```

$$(x - (-1)^{(1/4)})\,(x + (-1)^{(3/4)})\,(x - (-1)^{(3/4)})\,(x + (-1)^{(1/4)})$$

```
> plot(f3,x=-2..2,y=-3..2);
```



PŘÍKLAD 4: Polynom $f(x) = x^8 + x^4 + 1$ **graficky znázorněte a proveďte jeho rozklady v různých oborech integrity**

```
> f4:=x^8+x^4+1;
```

$$f4 := x^8 + x^4 + 1$$

```
> plot(f4,x=-1/2..1/2);
```



Příkaz **factor**, jehož použití se nabízí, ne vždy provede rozklad v symbolickém režimu. Jak vidíme dále, výsledkem rozkladu nad tělesem reálných či komplexních čísel jsou činitele, jejichž koeficienty jsou vyjádřeny jenom přibližně.

```
> Rozklad_Q:=factor(f4);
```

$$Rozklad\_Q := (x^2 + x + 1)\,(x^2 - x + 1)\,(x^4 - x^2 + 1)$$

```
> factor(f4,real);
```

$$(x^2 + 1.732050808\,x + 1.000000000)\,(x^2 + 1.000000000\,x + 1.000000000)\,(x^2 - 1.000000000\,x + 1.000000000)\,(x^2 - 1.732050808\,x + 1.000000000)$$

```
> factor(f4,complex);
```

$$(x + 0.8660254038 + 0.5000000000\,I)\,(x + 0.8660254038 - 0.5000000000\,I)\,(x + 0.5000000000 + 0.8660254038\,I)\,(x + 0.5000000000 - 0.8660254038\,I)$$

$$(x - 0.5000000000 + 0.8660254038\, I)\,(x - 0.5000000000 - 0.8660254038\, I)$$
$$(x - 0.8660254038 + 0.5000000000\, I)\,(x - 0.8660254038 - 0.5000000000\, I)$$

Pokud chceme rozklad v R[x] v uzavřeném tvaru, je třeba druhým parametrem příkazu factor specifikovat těleso koeficientů ireducibilních činitelů.

```
> Rozklad_R:=factor(f4,sqrt(3));
```

$$Rozklad\_R := -(x^2 + x\sqrt{3} + 1)\,(-x^2 + x\sqrt{3} - 1)\,(x^2 - x + 1)\,(x^2 + x + 1)$$

Symbolického vyjádření ireducibilních činitelů a úplného rozkladu na lineární činitele v C[x] dosáhneme následující kombinací příkazů **polytools[split]** a **convert**.

```
> Roz_f4:=polytools[split](f4,x);
```

$$Roz\_f4 := (1 + \mathrm{RootOf}(\_Z^2 + \_Z + 1) + x)\,(x - \mathrm{RootOf}(\_Z^2 - 1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)))$$
$$(x - 1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1))\,(x + \mathrm{RootOf}(\_Z^2 + \_Z + 1))$$
$$(x + \mathrm{RootOf}(\_Z^2 + \_Z + 1)\,\mathrm{RootOf}(\_Z^2 - 1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)))$$
$$(x - \mathrm{RootOf}(\_Z^2 + \_Z + 1))$$
$$(x - \mathrm{RootOf}(\_Z^2 + \_Z + 1)\,\mathrm{RootOf}(\_Z^2 - 1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)))$$
$$(x + \mathrm{RootOf}(\_Z^2 - 1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)))$$

```
> Rozklad_C:=convert(Roz_f4,radical);
```

$$Rozklad\_C := \left(\frac{1}{2} + \frac{1}{2} I\sqrt{3} + x\right)\left(x - \sqrt{\frac{1}{2} + \frac{1}{2} I\sqrt{3}}\right)\left(x - \frac{1}{2} - \frac{1}{2} I\sqrt{3}\right)\left(x - \frac{1}{2} + \frac{1}{2} I\sqrt{3}\right)$$
$$\left(x + \left(-\frac{1}{2} + \frac{1}{2} I\sqrt{3}\right)\sqrt{\frac{1}{2} + \frac{1}{2} I\sqrt{3}}\right)\left(x + \frac{1}{2} - \frac{1}{2} I\sqrt{3}\right)\left(x - \left(-\frac{1}{2} + \frac{1}{2} I\sqrt{3}\right)\sqrt{\frac{1}{2} + \frac{1}{2} I\sqrt{3}}\right)$$
$$\left(x + \sqrt{\frac{1}{2} + \frac{1}{2} I\sqrt{3}}\right)$$

Opakem uvedených příkazů pro rozklad na součin činitelů je příkaz **expand** pro roznásobení součinu (Jeho výsledek ještě zjednodušíme příkazem **simplify**).

```
> simplify(expand(Rozklad_C));
```

$$x^8 + x^4 + 1$$

PŘÍKLAD 5: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $x^4 + 4$

```
> restart;
> f5:=x^4+4;
```

$$f5 := x^4 + 4$$

```
> factor(f5);
```

$$(x^2 - 2\,x + 2)\,(x^2 + 2\,x + 2)$$

```
> factor(f5,complex);
```

$$(x + 1. + 1.\, I)\,(x + 1. - 1.\, I)\,(x - 1. + 1.\, I)\,(x - 1. - 1.\, I)$$

```
> Rozklad_f5:=polytools[split](f5,x);
```

$$Rozklad\_f5 := (x + \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2))\,(2 + \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2) + x)$$

$$(x-2-\mathrm{RootOf}(\_Z^2+2\,\_Z+2))\,(x-\mathrm{RootOf}(\_Z^2+2\,\_Z+2))$$

```
> convert(Rozklad_f5,radical);
```

$$(x-1+I)\,(1+I+x)\,(x-1-I)\,(x+1-I)$$

```
> plot(f5,x=-3..3,y=0..20);
```



PŘÍKLAD 6: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen x^6-1

```
> restart;
> f6:=x^6-1;
```

$$f6 := x^6-1$$

```
> factor(f6);
```

$$(x-1)\,(x+1)\,(x^2+x+1)\,(x^2-x+1)$$

```
> factor(f6,complex);
```

$$(x+1.000000000)\,(x+0.5000000000+0.8660254038\,I)$$
$$(x+0.5000000000-0.8660254038\,I)\,(x-0.5000000000+0.8660254038\,I)$$
$$(x-0.5000000000-0.8660254038\,I)\,(x-1.000000000)$$

```
> Rozklad_f6:=polytools[split](f6,x);
```

$$Rozklad\_f6 := (x+1)\,(x-1)\,(x-\mathrm{RootOf}(\_Z^2+\_Z+1))\,(1+\mathrm{RootOf}(\_Z^2+\_Z+1)+x)$$
$$(x-1-\mathrm{RootOf}(\_Z^2+\_Z+1))\,(x+\mathrm{RootOf}(\_Z^2+\_Z+1))$$

```
> convert(Rozklad_f6,radical);
```

$$(x+1)\,(x-1)\left(x+\frac{1}{2}-\frac{1}{2}I\sqrt{3}\right)\left(\frac{1}{2}+\frac{1}{2}I\sqrt{3}+x\right)\left(x-\frac{1}{2}-\frac{1}{2}I\sqrt{3}\right)\left(x-\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)$$

```
> plot(f6,x=-3..3,y=-2..10);
```



PŘÍKLAD 7: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $x^3-6\,x^2+11\,x-6$

```
> restart;
> f7:=x^3-6*x^2+11*x-6;
```

$$f7 := x^3 - 6\,x^2 + 11\,x - 6$$

```
> factor(f7);
```

$$(x-1)\,(x-2)\,(x-3)$$

```
> factor(f7,complex);
```

$$(x-1.000000000)\,(x-2.000000000)\,(x-3.000000000)$$

```
> Rozklad_f7:=polytools[split](f7,x);
```

$$Rozklad\_f7 := (x-1)\,(x-2)\,(x-3)$$

```
> plot(f7,x=-1..5,y=-2..6);
```



PŘÍKLAD 8: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen
$x^3 - 4\,x^2 + 4\,x - 3$

```
> restart;
> f8:=x^3-4*x^2+4*x-3;
```

$$f8 := x^3 - 4\,x^2 + 4\,x - 3$$

```
> factor(f8);
```

$$(x-3)\,(x^2 - x + 1)$$

```
> factor(f8,complex);
```

$$(x - 0.5000000000 + 0.8660254038\,I)\,(x - 0.5000000000 - 0.8660254038\,I)\,(x - 3.)$$

```
> Rozklad_f8:=polytools[split](f8,x);
```

$$Rozklad\_f8 := (-1 + \mathrm{RootOf}(\_Z^2 - \_Z + 1) + x)\,(x-3)\,(x - \mathrm{RootOf}(\_Z^2 - \_Z + 1))$$

```
> convert(Rozklad_f8,radical);
```

$$\left(-\frac{1}{2} + \frac{1}{2}\,I\,\sqrt{3} + x\right)(x-3)\left(x - \frac{1}{2} - \frac{1}{2}\,I\,\sqrt{3}\right)$$

```
> plot(f8,x=-5..5,y=-5..5);
```



PŘÍKLAD 9: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $16\,x^4 + 64$

```
> restart;
> f9:=16*x^4+64;
```

$$f9 := 16\,x^4 + 64$$

```
> factor(f9);
```

$$16\,(x^2 - 2\,x + 2)\,(x^2 + 2\,x + 2)$$

```
> factor(f9,complex);
```

$$16.\,(x + 1. + 1.\,I)\,(x + 1. - 1.\,I)\,(x - 1. + 1.\,I)\,(x - 1. - 1.\,I)$$

```
> Rozklad_f9:=polytools[split](f9,x);
```

$$Rozklad\_f9 := 16\,(2 + \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2) + x)\,(x - \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2))$$
$$(x - 2 - \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2))\,(x + \mathrm{RootOf}(\_Z^2 + 2\,\_Z + 2))$$

```
> convert(Rozklad_f9,radical);
```

$$16\,(1 + I + x)\,(x + 1 - I)\,(x - 1 - I)\,(x - 1 + I)$$

```
> plot(f9,x=-5..5,y=-0..250);
```




```
>
```